
	Excel函数有很多，很多函数用法小编都有给大家分享过，不过今天要给大家重磅介绍的是Vlookup函数最经典的10种用法，掌握它们，说不定工作上有意外惊喜哦!下面就一起来看下吧!



	1、计算个人所得税



	=(G28-3500)*VLOOKUP(G28-3500,C28:E34,2)-VLOOKUP(G28-3500,C28:E34,3)



	公式说明：当Vlookup最后一个参数为1或省略时，查找为近似匹配。即查找比指定值小且最接近的值。>>>推荐阅读：2018新旧个税差异到底在哪里?<<<

	2、查找个人信息



	=VLOOKUP($B18,$C$7:$H$15,COLUMN(B1),0)



	注：Colum(b1)的结果是2，当公式向右复制时可以生成3，4，5，..

	3、合并两个表格



	F2单元格公式



	=VLOOKUP(B2,$B$14:$F$22,MATCH(F$1,$B$13:$F$13,0))



	公式说明：用Match函数查找表1的标题在表2中的位置

	4、让两个表格同类商品排序完全一样



	F2单元格公式



	=IFERROR(VLOOKUP($B2,$B$12:$D$16,COLUMN(A1),),"")



	公式说明：IFERROR函数用来屏蔽错误值

	5、模糊查找产品名称(包含查找)



	=VLOOKUP("*"&F41&"*",$B$41:$D$47,3,0)



	注：*是通配符，代表任意多个字符

	6、带特殊符合的字符查找



	=VLOOKUP(SUBSTITUTE(F52,"~","~~"),B$52:C$56,2,0)



	注：由于~ * ?在公式中有通配符用法，遇到被查找的字符串中含有这3个字符就用Substitute替换为~~、~*和~?

	7、逆向查找



	=VLOOKUP(G61,IF({1,0},C61:C69,B61:B69),2,)



	公式说明: 本公式只做了解，遇到逆向查找，建议使用Index+Match函数。



	=INDEX(B60:B69,MATCH(G61,C60:C69,0))

	8、Vlookup多条件查找



	A112 =B112&C112



	H112 =VLOOKUP(F112&G112,A112:D120,4,0)



	公式说明：Vlookup也可以使用IF({1,0}的数组形式实现多条件查找，但运行效率太慢，所以建议使用辅助列连接再查找的方法。

	9、16年与17年单价核对，表格核对(带通配符)，结果显示多N或少N



	=TEXT(VLOOKUP(SUBSTITUTE(B73,"*","~*"),$F$73:$G$92,2,)-C73,"少0.00;多0.00;")



	公式说明:Text可以设置数字显示的格式

	10、库存单价的监控。销售单价>库存单价时填充红色背景

	设置方法：选取区域 - 条件格式 - 新建规则 - 使用公式确定... - 在下面框中输入公式：



	=H2>VLOOKUP(F2,A:C,3,0)



	最后点“格式”按钮设置填充色为红色

		以上就是“Vlookup函数最经典的10种用法”全部介绍，不知道小编分享完，大家有没有额外收获呢?>>>推荐阅读：vlookup函数常见的10错误<<<
	



	欢迎关注PDF帝，更多精彩内容持续更新中!


				

				

			

						
					最新文章
	
							                    	他们都不知道PPT、PDF、Word这3个互相转换的方法
	                    	                    	2018新旧个税差异到底在哪里?
	                    	                    	2018年个人所得税新算法
	                    	                    	Excel八大经典函数组合整理篇
	                    	                    	神奇的Excel组合键:Ctrl+ [
	                    	                    	Sumif函数的一个奇怪用法
	                    	                    	你会用Excel表格做"国旗"吗?
	                    	                    	多篇Word文档合并技巧
	                    	                    	撤销终于不是用Ctrl+Z,而是用用Ctrl键了...
	                    	                    	大家用过Ctrl+K吗?好用吗?
	                    					


			

						分享到：QQ空间新浪微博腾讯微博人人网微信


上一篇：PDF水印如何批量删除？批量去PDF文档水印教程

下一篇：Vlookup+Vlookup更牛了！

